TP 2 : Des suites et des intégrales

```
> restart;
```

1 Des récurrences du premier ordre non monotone

1. u[n+1]=exp(-u[n])

```
> f:=x->exp(-x):plot({f(x),x},x=0..1);
```




```
> fsolve(f(x)=x);
```

$$.5671432904$$

```
> u:=n->if n=0 then 1. else f(u(n-1)) fi:
> seq(u(k),k=0..20);
```

1., .3678794412, .6922006275, .5004735006, .6062435351, .5453957860, .5796123355, .5601154614, .5711431151, .5648793474, .5684287250, .5664147332, .5675566373, .5669089119, .5672762322, .5670678984, .5671860501, .5671190401, .5671570440, .5671354902, .5671477143

Les suites extraites d'indices pairs et impairs semblent adjacentes.
Déjà, elles vérifient des relations de la forme v[n+1]=f(f(v[n])), et fof est strictement croissante, donc la relation v[1]>v[0] se propage : v est donc strictement croissante, et de meme w est strictement décroissante. Elles sont par ailleurs localisées dans [0,1], d'où les convergences. Leurs limites l1 et l2 sont solutions de l'équation f(f(x))=x. Bien entendu, l'unique X tel que f(X)=X vérifie également f(f(X))=X, mais il pourrait y en avoir d'autres, a priori.

```
> plot({f(f(x)),x},x=0..1);
```



Il semblerait qu'il n'y ait qu'un point fixe. Pour cela, on considère la différence f(f(x))-x

```
> diff(f(f(x))-x,x);
```

$$\mathbf{e}^{(-x)}\,\mathbf{e}^{(-\mathbf{e}^{(-x)})}-1$$

g:x->f(f(x))-x est strictement décroissante, et g(0)>0>g(1), donc il existe un unique point l tel que g(l)=0, c'est-à-dire f(f(l))=l; gagné.

2. u[n+1]=exp(-3u[n])

```
> g:=x->exp(-3*x):plot({g(x),x},x=0..1);
```




```
> fsolve(g(x)=x);
```

.3499696317

```
> v:=n->if n=0 then 1. else g(v(n-1)) fi:
> seq(v(k),k=0..20);
```

1., .04978706837, .8612579679, .07548857913, .7973466610, .09144295141, .7600820780, .1022590240, .7358146163, .1099814186, .7189638104, .1156841749, .7067682045, .1199950708, .6976866431, .1233092404, .6907842579, .1258892441, .6854582185, .1279168730, .6813013096

Les suites extraites d'indices pairs et impairs semblent respectivement croissante et décroissante (on le prouverait comme pour u) et localisées dans [0,1], d'où les convergences. Par contre, il semblerait que leurs limites l1 et l2 soient différentes. Elles sont solutions de l'équation g(g(x))=x. Ici encore, l'unique X tel que g(X)=X vérifie également g(g(X))=X, mais on va voir qu'il y a d'autres solutions.

```
> plot({g(g(x)),x},x=0..1);
```




```
> plot(g(g(x))-x,x=0..1);
```



C'est plus clair, non ?
Il semblerait qu'il n'y ait qu'un point fixe. Pour cela, on considère la différence f(f(x))-x

```
> fsolve(g(g(x))=x);
```

.1361198833

```
> fsolve(g(g(x))=x,x=.2..0.5);
```

.3499696317

```
> fsolve(g(g(x))=x,x=.5..1);
```

.6647397623

Une étude de x->gog(x)-x etablirait effectivement l'existence de trois solutions pour l'équation g(g(l))=l. Si on note x1,x2,x3 ces solutions (dans l'ordre croissant), on montrerait que l'intervalle [0,x1] est stable par gog, donc tous les termes pairs de v sont dans cet intervalle, donc majorés par x1, donc leur limite également, donc cette limite EST x1. De meme, on montrerait que les termes impairs tendent vers x3.

2 Des exercices du TD d'intégration

1

```
> limit(sum(1/(n^4+k^2*n^2+k^4),k=1..n),n=infinity);
```

$$0$$

```
> u:=n->n^3*add(evalf(1/(n^4+k^2*n^2+k^4)),k=1..n):
> seq(u(10**k),k=0..5);
```

.3333333333, .6942140265, .7247640237, .7277695241, .7280695830, .7280995830

```
> convert(1/(1+x^2+x^4),parfrac,x);
```

$$-\frac{1}{2}\,\frac{-1+x}{x^2-x+1}+\frac{\frac{1}{2}(1+x)}{1+x+x^2}$$

```
> int(1/(1+x^2+x^4),x=0..1);evalf(%);
```

$$\frac{1}{12}\pi\sqrt{3}+\frac{1}{4}\ln(3)$$

$$.7281029134$$

Cette intégrale est en fait la limite recherchée. La première réponse de Maple est donc fausse...

2

```
> asympt(sum(1/k,k=1..n),n);
```

$$\ln(n)+\gamma+\frac{\frac{1}{2}}{n}-\frac{1}{12}\,\frac{1}{n^2}+\frac{\frac{1}{120}}{n^4}+O\left(\frac{1}{n^6}\right)$$

```
> asympt(sum(k*ln(k),k=1..n),n);
```

$$\left(-\frac{1}{4}+\frac{1}{2}\ln(n)\right)n^2+\frac{1}{2}\ln(n)\,n-\zeta(1,\text{-}1)+\frac{1}{12}+\frac{1}{12}\ln(n)+\frac{\frac{1}{720}}{n^2}+O\left(\frac{1}{n^4}\right)$$

3

```
> u:=(n,p)->(sum((1+k/n)^(1/p),k=0..n-1)/n)^p:
> limit(u(n,p),n=infinity);
```

$$\left(\frac{p\left(2^{\left(\frac{1+p}{p}\right)}-1\right)}{1+p}\right)^p$$

```
> limit(%,p=infinity);
```

$$4\,\mathbf{e}^{(-1)}$$

```
> limit(u(n,p),p=infinity);
```

$$e^{\left(1/2\,\frac{2\,n\ln\left(\frac{1}{n}\right)-2\ln(2)-\ln(\pi)+2\ln\left(\left(2^{n}\right)^{2}\Gamma(n)\,\Gamma(n+1/2)\right)-2\ln(\Gamma(n))}{n}\right)}$$

```
> limit(%,n=infinity);
```

$$4\,e^{(-1)}$$

4

```
> int(cos(x)/x,x=u..2*u);
```

$$\mathrm{Ci}(2\,u)-\mathrm{Ci}(u)$$

Héhé !

```
> ?Ci
Ci(x)   = gamma + ln(x) + int((cos(t)-1)/t, t=0..x)
> plot(Ci,0..1);
```




```
> limit(Ci(2*u)-Ci(u),u=0);
```

$$\ln(2)$$

5

```
> f:=x->int(sqrt(1-t^8),t=x^2/6..3*x^3);
```

$$f := x \to \int_{1/6\,x^2}^{3\,x^3} \sqrt{1-t^8}\,dt$$

```
> plot(f,-3^(-1/3)..3^(-1/3));
```




```
> D(f);
```

$$x \to 9\,x^2\sqrt{1-6561\,x^{24}}-\frac{1}{3}\,x\sqrt{1-\frac{1}{1679616}\,x^{16}}$$

```
> simplify(diff(f(x),x));
```

$$\frac{1}{14171760}\,x\Bigg(-836828256240\,x^{25}\sqrt{1679616-x^{16}}+$$

$$297538935552\,x^{25}\sqrt{1-6561\,x^{24}}\sqrt{1679616-x^{16}}\ \mathrm{hypergeom}\left(\left[\frac{3}{2},\frac{9}{8}\right],\left[\frac{17}{8}\right],6561\,x^{24}\right)$$

$$+3645\sqrt{1-6561\,x^{24}}\,x^{16}$$

$$-x^{16}\sqrt{1-6561\,x^{24}}\sqrt{1679616-x^{16}}\ \mathrm{hypergeom}\left(\left[\frac{3}{2},\frac{9}{8}\right],\left[\frac{17}{8}\right],\frac{1}{1679616}\,x^{16}\right)$$

$$+\,102036672\,x\,\mathrm{hypergeom}\left(\left[\frac{1}{8},\frac{1}{2}\right],\left[\frac{9}{8}\right],6561\,x^{24}\right)\sqrt{1-6561\,x^{24}}\,\sqrt{1679616-x^{16}}$$
$$+\,25509168\,x\sqrt{1679616-x^{16}}$$
$$-\,3779136\,\mathrm{hypergeom}\left(\left[\frac{1}{8},\frac{1}{2}\right],\left[\frac{9}{8}\right],\frac{1}{1679616}\,x^{16}\right)\sqrt{1-6561\,x^{24}}\,\sqrt{1679616-x^{16}}$$
$$-\,1224440064\sqrt{1-6561\,x^{24}}\Bigg)\Big/\left(\sqrt{1-6561\,x^{24}}\,\sqrt{1679616-x^{16}}\right)$$

Ha ha !

3 Des calculs d'intégrales

```
> int(1/(x*(x^2+5)),x=1..2);
```

$$\frac{3}{10}\ln(2)-\frac{1}{10}\ln(3)$$

```
> int(1/(x^2-4),x=-1..1);
```

$$-\frac{1}{2}\ln(3)$$

```
> int((2*x+3)/(2*x+1),x=0..1);
```

$$1+\ln(3)$$

```
> int((x^2-5*x+5)/(x^2+4),x=0..1);
```

$$\frac{1}{2}\arctan\left(\frac{1}{2}\right)+1-\frac{5}{2}\ln(5)+5\ln(2)$$

```
> int(1/(x^3+1),x=0..1);
```

$$\frac{1}{3}\ln(2)+\frac{1}{9}\pi\sqrt{3}$$

```
> int(x/(x+1)^2,x=0..t);
```

$$\frac{\ln(t+1)\,t+\ln(t+1)-t}{t+1}$$

Avec IPP

```
> int(x^2*(2*x+1)^10,x=0..1);
```

$$\frac{2690420}{429}$$

```
> ?intparts
```

student[intparts] - perform integration by parts

Calling Sequence

intparts(f, u)

Parameters

`f` - an expression of the form Int(u*dv, x)
`u` - the factor of the integrand to be differentiated

Description

- Carry out integration by parts on an unevaluated integral (expressed in terms of Int).
- The function returns u*v - Int(du*v, x) as its value.

```
> with(student):
> intparts(Int(x^2*(2*x+1)^10,x=0..1),x^2);
```

$$\frac{177147}{22} - \int_0^1 \frac{1}{11}\, x\, (2\,x+1)^{11}\, dx$$

```
> intparts(%,x);
```

$$\frac{531441}{88} + \int_0^1 \frac{1}{264}\, (2\,x+1)^{12}\, dx$$

```
> value(%);
```

$$\frac{2690420}{429}$$

```
> int(x^19*(x^2+1)^18,x=-1..1);
```

$$0$$

```
> int(sinh(t)/(1+cosh(t)),t=a..b);
```

$$\ln(1+\cosh(b)) - \ln(1+\cosh(a))$$

```
> int(x*arctan(x)^2,x=0..1);
```

$$-\frac{1}{4}\ln(1-I)^2 + \frac{1}{2}\ln(1-I)\ln(1+I) - \frac{1}{2}I\ln(1-I) - \frac{1}{4}\ln(1+I)^2 + \frac{1}{2}I\ln(1+I) + \frac{1}{2}\ln(2)$$

Arf ! Cela n'a pas de sens pour vous (ln d'un complexe... aheum...)

```
> evalc(%);
```

$$\frac{1}{16}\pi^2 - \frac{1}{4}\pi + \frac{1}{2}\ln(2)$$

Mieux...

```
> int(cos(2*x)^2/sin(x),x=Pi/2..alpha);
```

$$\frac{4}{3}\cos(\alpha)^3 + \ln(\csc(\alpha) - \cot(\alpha))$$

```
> simplify(%);
```

$$\frac{4}{3}\cos(\alpha)^3 + \ln\left(\frac{\sin(\alpha)}{\cos(\alpha)+1}\right)$$

```
> int((1+cosh(x))/(1+sinh(x)^2),x=0..1);
```

$$-\frac{\mathbf{e}^{(-1)} + 2\arctan\left(\frac{\mathbf{e}^{(-1/2)} - \mathbf{e}^{(1/2)}}{\mathbf{e}^{(-1/2)} + \mathbf{e}^{(1/2)}}\right)\mathbf{e} - \mathbf{e} + 2\arctan\left(\frac{\mathbf{e}^{(-1/2)} - \mathbf{e}^{(1/2)}}{\mathbf{e}^{(-1/2)} + \mathbf{e}^{(1/2)}}\right)\mathbf{e}^{(-1)}}{\mathbf{e}^{(-1)} + \mathbf{e}}$$

```
> simplify(%);
```

$$\frac{-1 + 2\arctan\left(\frac{-1+\mathbf{e}}{1+\mathbf{e}}\right)\mathbf{e}^2 + \mathbf{e}^2 + 2\arctan\left(\frac{-1+\mathbf{e}}{1+\mathbf{e}}\right)}{1+\mathbf{e}^2}$$

```
> int(cos(t)^3,t=0..Pi/2);
```

$$\frac{2}{3}$$

```
> int(sin(t)^5,t=0..Pi/2);
```

$$\frac{8}{15}$$

```
> int(sqrt(1+abs(x*(1-x))),x=-1..1);
```

$$\frac{1}{4}+\frac{5}{4}\arcsin\left(\frac{1}{5}\sqrt{5}\right)-\frac{3}{16}\ln(3)+\frac{3}{4}\sqrt{3}-\frac{3}{8}\ln(2-\sqrt{3})$$

```
> int((cos(x))/(1+sin(x)^3),x=0..Pi/2);
```

$$\frac{1}{3}\ln(2)+\frac{1}{9}\sqrt{3}\,\pi$$

4 Une étude de fonction

```
> plot(exp(t^2),t=-1..1);
```




```
> solve(int(exp(t^2),t=x..y)=1,y);
```

$$-I\,\mathrm{RootOf}(-\mathrm{erf}(\_Z)\sqrt{\pi}+\sqrt{\pi}\,\mathrm{erf}(I\,x)+2\,I)$$

Arf !

```
> f:=x->solve(int(exp(t^2),t=x..y)=1,y):
> plot({x,f(x)},x=-2..2);
```




```
> diff(f(x),x);;
```

$$\frac{\mathbf{e}^{(x^2)}}{\mathbf{e}^{(-\mathrm{RootOf}(-\mathrm{erf}(\_Z)\sqrt{\pi}+\sqrt{\pi}\,\mathrm{erf}(I\,x)+2\,I)^2)}}$$

Mouais...

```
> asympt(f(x),x);
Error, (in asympt) unable to compute series
```

Tout de meme...

```
> simplify(f(-f(x)));
```

$$-I\,\mathrm{RootOf}(-\mathrm{erf}(\_Z)\sqrt{\pi}-\sqrt{\pi}\,\mathrm{erf}(\mathrm{RootOf}(-\mathrm{erf}(\_Z)\sqrt{\pi}+\sqrt{\pi}\,\mathrm{erf}(I\,x)+2\,I))+2\,I)$$

```
> plot(f(-f(x)),x=-1..1);
```



On peut effectivement montrer que f(-f(x))=-x. Cela montre que si (x,y) est sur le graphe de f, alors (-y,-x) l'est aussi : le graphe de f présente donc une symétrie par rapport à la droite d'équation y=-x.